

EE324 Control Systems Lab

Problem Sheet 5

Prakhar Diwan, Roll no. 180100083

1. Problem 1

1.1. 1a

We are given $H_{closed}(s)$ the closed loop transfer function of a system with unity negative feedback. Thus we find the open loop transfer function $G_{open}(s)$ as:

$$G_{open}(s) = \frac{H_{closed}(s)}{1 - H_{closed}(s)} = \frac{10}{s^3 + 4s^2 + 5s}$$

The root-locus is plotted in Figure 1.




Figure 1: Root-Locus for 1a

We observe there are 3 poles, and the system's poles lie in the LHP till about Critical gain = 2. Beyond that, the system becomes unstable.

```
clear;
clc;
s = poly(0, 's');
H = 10/(s^3 + 4*s^2 + 5*s + 10); //closed loop transfer func
G_open = H/(1-H); // applying a positive unity feedback
//for getting open loop transfer fn
poles = roots(G_open.den);
```

```
G_open = syslin('c',G_open);
evans(G_open); // constructing root locus
```

1.2. 1b

We are given $G_{open}(s)$ the open loop transfer function as:

$$G_{open}(s) = \frac{s+1}{s^2(s+3.6)}$$

The root-locus is plotted in Figure 2.




Figure 2: Root-Locus for 1b

We observe there are 3 poles (including 1 pair of repeated poles at $s = 0$). We have 2 asymptotes (as the $n_{poles} - n_{zeros} = 2$)

```
clear;
clc;
s = poly(0, 's');
G = (s+1)/((s^2)*(s+3.6));
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G); // constructing root locus
```

1.3. 1c

We are given $G_{open}(s)$ the open loop transfer function as:

$$G_{open}(s) = \frac{s + 0.4}{s^2(s + 3.6)}$$

The root-locus is plotted in Figure 3.




Figure 3: Root Locus for 1c

```
clear;
clc;
s = poly(0, 's');
G = (s+0.4)/((s^2)*(s+3.6));
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G,50); // constructing root locus
```

1.4. 1d

We are given $G_{open}(s)$ the open loop transfer function as:

$$G_{open}(s) = \frac{s + p}{s(s + 1)(s + 2)}$$

The value of p was varied and the following observations were made:

For $p < 0$, the system is bound to be unstable as one of the open loop poles will try to reach this open loop zero (lying in ORHP), implying a branch being in ORHP. Therefore, the system being unstable.

For $p = 0, 1$ and 2 , the system is stable due to the reduction to common factors cancelling out, and the system being reduced to a second order system with zeros at infinity and break-away point situated on negative real axis. One of the plots, for $p = 0$, is shown below:




Figure 4: Root Locus for $p = 0$

For $0 < p < 2$, I observed that the system was stable for all value of open loop gain ($K > 0$). Root locus for one of the values ($p = 1.5$) is plotted below:




Figure 5: Root Locus for $p = 1.5$

For $2 < p < 3$ as well the system was observed to be stable for all values of open loop gain ($K > 0$). Root locus plot for $p = 2.5$ is given below:




Figure 6: Root Locus for $p = 2.5$

For $p = 3$, the critical point is reached for system's stability as at $p = 3$, the asymptotes coincide with the imaginary axis, thereby the system being marginally stable as K tends to infinity. Root locus plot for $p = 3$, is shown below:




Figure 7: Root Locus for $p = 3$

For $p > 3$, we observe that the 2 asymptotes are situated in ORHP, implying the system being stable only for some values of open loop gain ($K > 0$), and unstable for $K > C$, where C is a constant which varies with p . Root locus plot for $p = 5$ is shown below:




Figure 8: Root Locus for $p = 5$

```
clear;
clc;
s = poly(0, 's');
p = 0;
G = (s+p)/(s*(s+1)*(s+2));
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G,1000); // constructing root locus
end
```

2. Problem 2

2.1. 2a

We need the break-in and break-away points to coincide. So we take the specific case of poles placed on the real axis symmetrically about the origin, and the zeros placed symmetrically about the origin on the imaginary axis.

$$G_{open}(s) = \frac{s^2 + 4}{s^2 - 4}$$




Figure 9: Root Locus for 2a

The point of coincidence (breakin and breakaway) is the origin.

```
clear;
clc;
s = poly(0, 's');
k = 0;
G = ((s+k)^2+4)/((s+k)^2-4);
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G); // constructing root locus
```

2.2. 2b

We consider the case of 5 branches at the breakaway point (s=0):

$$G_{open}(s) = \frac{s^5 + 1}{s^5 - 1}$$




Figure 10: Root locus for 2b

By placing the poles at points that divide the circumference of the unit circle around the origin into 5 equal parts (and the zeros at five roots of unity as depicted by the numerator), we obtain the required condition. The point of coincidence (breakin and breakaway) is the origin.

```
clear;
clc;
s = poly(0, 's');
G = (s^5+1)/(s^5-1);
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G,100); // constructing root locus
```

2.3. 2c

We can obtain a Root locus with the branches coinciding with the asymptotes by placing all the poles at the location where the asymptotes seem to meet, and the zeros at infinity (basically no zeros).

$$G_{open}(s) = \frac{1}{(s+1)^5}$$




Figure 11: Root locus for 2c

The point of coincidence (breakin and breakaway) is the point $s = -1$.

```
clear;
clc;
s = poly(0, 's');
G = 1/((s+1)^5);
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G,100); // constructing root locus
```

2.4. 2d

(i) We consider the transfer function $G_1(s)$, with poles placed on the real axis, symmetrically about the $j\omega$ axis:

$$G_1(s) = \frac{1}{(s^2 - 1)(s^2 - 4)} = \frac{1}{s^4 - 5s^2 + 4}$$




Figure 12: Root locus for $G_1(s)$

(ii) Obtain $G_2(s)$ by substituting s^2 with $-s^2$

$$G_2(s) = \frac{1}{(s^2 + 1)(s^2 + 4)}$$




Figure 13: Root locus for $G_2(s)$

(iii) Finally replace s with $s - 4$:

$$G_3(s) = \frac{1}{((s-4)^2 + 1)((s-4)^2 + 4)}$$




Figure 14: Root locus for $G_3(s)$

Thus we have obtained complex break-out points.

```
clear;
clc;
s = poly(0, 's');
G = 1/((s^2-1)*(s^2-4));
G = syslin('c',G);
poles = roots(G.den); //open loop poles
// evans finds root-locii of 1+K*G(s)
evans(G,100); // constructing root locus

G_2 = 1/((s^2+1)*(s^2+4));
G_2 = syslin('c',G_2);
poles = roots(G_2.den); //open loop poles
// evans finds root-locii of 1+K*G_2(s)
evans(G_2,100); // constructinG_2 root locus

k = 4;
G_3 = 1/(((s-k)^2+1)*((s-k)^2+4));
G_3 = syslin('c',G_3);
poles = roots(G_3.den); //open loop poles
```

```
// evans finds root-locii of 1+K*G_3(s)
evans(G_3,100); // constructinG_3 root locus
```

3. Problem 3

$$G(s) = \frac{1}{s(s^2 + 3s + 5)}$$

Root Locus for G(s) is plotted below:




Figure 15: Root locus for G(s)

The system is stable for $kp \leq 15$, as was observed from the above RL plot.

Upon plotting for values of kp from 0.1 to 15, and fine tuning for later simulations, $kp = 3.7475$ was found to be optimal value for rise time of 1.5s.

For a stable system, the highest value of $kp = 15$ leads to lowest rise time (0.57s). If kp value is further increased, rise time further decreases.

```
clear;
clc;
s=poly(0,'s');
G = syslin('c',1 / (s*(s^2 + 3*s +5)));
evans(G);

k_max = 15;
t=0:0.01:15;
kp = 0.1:0.1:15;
//kp=3.7:0.001:3.8;
for k1=kp
    G1 = (k1/(k1+s*(s^2 + 3*s +5)));
    G = syslin('c',G1);
```

```

G_rep = csim('step',t,G);
if(max(G_rep)>=0.9) then
    tr = t(find(G_rep > 0.9)(1))-t(find(G_rep > 0.1)(1));
    printf("k=%.3f , rise_time = %.3f\n",k1,tr);
end;
end;

```

4. Problem 4

Open loop transfer function:

$$G(s) = \frac{0.11(s + 0.6)}{6s^2 + 3.6127s + 0.0572}$$

The root locus plot for the $G(s)$ as open loop transfer function is given below:




Figure 16: Root locus for $kp > 0$

For 1% steadystate error on step input $\Rightarrow \lim_{s \rightarrow 0} \frac{1}{1 + kpG(s)} = 0.01$

$$kp = \frac{99 \times 0.0572}{0.11 \times 0.6} = 85.8$$

Using Scilab, $kp = 85.8$ gave steady state error = 1%. The unit response is shown below:




Figure 17: Unit step response for $kp = 85.8$

For $kp > 0$, system is stable as depicted by evans plot. For $kp < 0$, i just multiplied G with -1 and used evans so Loop gain marked in the RL plot below corresponds to $kp = -0.8659$.




Figure 18: Root locus for $kp < 0$

The system is hence observed to be marginally stable for $kp = -0.867$.

```
clear;
clc;
s=poly(0,'s');
G = ((0.11*(s+0.6))/(6*s^2 + 3.6127*s + 0.0572));
G = syslin('c', G);
evans(G,100);
```

```

t=0:0.01:20;
kp = 85.8;
G1 = (kp*G/(1+kp*G));
H = syslin('c',G1);
H_rep = csim('step',t,H);
plot(t,H_rep);
var = gca();
var.data_bounds = [0,0; 20,1.2];
xtitle ('Unit step response for kp = 85.8','time instants' , 'unit step response' );

```

5. Problem 5

$$G_1(s) = \frac{25}{(s+50)(s^2+s+0.5)}$$

$$G_2(s) = \frac{0.5}{s^2+s+0.5}$$




Figure 19: Root Locus for $G_1(s)$




Figure 20: Root Locus for $G_2(s)$

As k_p increases, the roots of the third order system move towards that RHP while the roots of the second order system have the same real part. Hence as k_p increases, the third order system experiences lesser rising time and peak time but more settling time and percentage overshoot; whereas the second order system has a constant settling time and its percentage overshoot increases and peak time decreases as k_p increases. The difference amongst the step responses becomes prominent around $k = 41$.




Figure 21: k = 1




Figure 22: k = 21




Figure 23: k = 41




Figure 24: k = 61




Figure 25: k = 81




Figure 26: k = 101

```
clear;
clc;
s=poly(0,'s');

G1 = (25/((s+50)*(s^2+s+0.5)));
G2 = 0.5/(s^2+s+0.5);
```

```
sysG1 = syslin('c',G1);
sysG2 = syslin('c',G2);

fig=scf();
evans(sysG1);
fig=scf();
evans(sysG2);

k = 1:20:101;
t=0:0.01:10;
for k1=k
    sysG1 = syslin('c',k1*G1/(1+k1*G1));
    sysG2 = syslin('c',k1*G2/(1+k1*G2));
    step1 = csim('step',t,sysG1);
    step2 = csim('step',t,sysG2);
    fig=scf();
    plot(t,step1,t,step2);
    title('k='+string(k1));
    h2=legend(['third order','second order'])
end;
```