EE 324: Assignment 1: Analysis in Laplace Domain

Prakhar Diwan (180100083)

- The principles used are mentioned in details as comments corresponding to the question in the Scilab code, which can be found at the end under appendix.

Question 1:

(a) After first stage, output is R(s)*G1(s) which is fed to G2. Thus finally, C(s)=G1(s)*G2(s)*R(s). Using Scilab, we get transfer function (C(s)/R(s)) as:

$$\frac{50}{50 + 20s + 7s^2 + s^3}$$

(b) Output from G1 (R(s)*G1(s)) is added to output from G2 (R(s)*G2(s)).
Thus, C(s) = (G1(s)+G2(s)) *R(s),
Using Scilab, we get transfer function (C(s)/R(s)) as

$$\frac{100 + 20s + 5s^2}{50 + 20s + 7s^2 + s^3}$$

(c) Upon relating the inputs and outputs in the Laplace domain, we get $C = (R - CG_2)G_1$, which simplifies to $\frac{C}{R} = \frac{G_1}{1+G_1G_2}$. Using Scilab, we get transfer function (C(s)/R(s)) as

$$\frac{50 +10s}{100 +20s +7s^2 +s^3}$$

(d) Plot of response to the unit step to the system with transfer function G1(s).



Question 2:

After finding the transfer functions corresponding to 1(a), 1(b) and 1(c), numerator and denominator of each transfer function are extracted using .num and .den functions. Then the roots of the numerator and the denominator are found using the roots() function (used to find roots of a polynomial), thus giving us zeros and poles for each of the transfer functions in 1(a), 1(b) and 1(c).

The poles and zeros found using Scilab are shown below:

Part	Zeros	Poles
1(a) Cascade System	No Zeros	-5 + 0*i, -1 + 3*i, -1 – 3*i
1(b) Parallel System	-2 + 4*i -2 – 4*i	-5 + 0*i, -1 + 3*i, -1 – 3*i
1(c) Feedback (closed loop) System	-5	-6.3347665 -0.3326167 + 3.9592004*i -0.3326167 - 3.9592004*i

Computations of matrix operations on polynomial matrices M1 and M2:

M1 =

```
                              2
    10               100 + 20s + 5s
------------        ------------------
             2                     2   3
10 + 2s + s         50 + 20s + 7s + s

       50                5
------------------  ------
             2   3
50 + 20s + 7s + s   5 + s
```

M2 =

```
     50 + 10s          s
-------------------   --
              2   3
100 + 20s + 7s + s    1

1 + s                   5
------                ------

  1                   5 + s
```

Addition of M1 and M2 =

$$\begin{bmatrix} \dfrac{1500 + 400s + 140s^2 + 20s^3}{1000 + 400s + 210s^2 + 44s^3 + 9s^4 + s^5} & \dfrac{100 + 70s + 25s^2 + 7s^3 + s^4}{50 + 20s + 7s^2 + s^3} \\[2ex] \dfrac{100 + 70s + 27s^2 + 8s^3 + s^4}{50 + 20s + 7s^2 + s^3} & \dfrac{10}{5 + s} \end{bmatrix}$$

Product of M1 and M2=

$$\begin{bmatrix} \dfrac{12500 + 15000s + 5700s^2 + 1940s^3 + 395s^4 + 60s^5 + 5s^6}{5000 + 3000s + 1450s^2 + 430s^3 + 89s^4 + 14s^5 + s^6} & \dfrac{500 + 350s + 125s^2 + 10s^3}{250 + 150s + 55s^2 + 12s^3 + s^4} \\[2ex] \dfrac{7500 + 7500s + 3050s^2 + 1270s^3 + 265s^4 + 50s^5 + 5s^6}{5000 + 3000s + 1450s^2 + 430s^3 + 89s^4 + 14s^5 + s^6} & \dfrac{250 + 300s + 75s^2}{250 + 150s + 55s^2 + 12s^3 + s^4} \end{bmatrix}$$

Determinant of M1=

$$\frac{-2500 + 100s^2 + 50s^3}{2500 + 2000s + 1100s^2 + 380s^3 + 89s^4 + 14s^5 + s^6}$$

Inverse of M1 =

$$\begin{bmatrix} \dfrac{2500 + 1500s + 800s^2 + 220s^3 + 45s^4 + 5s^5}{-2500 + 100s^2 + 50s^3} & \dfrac{-5000 - 3000s - 1350s^2 - 340s^3 - 55s^4 - 5s^5}{-2500 + 100s^2 + 50s^3} \\[2ex] \dfrac{-2500 - 1000s - 350s^2 - 50s^3}{-2500 + 100s^2 + 50s^3} & \dfrac{2500 + 1500s + 550s^2 + 120s^3 + 10s^4}{-2500 + 100s^2 + 50s^3} \end{bmatrix}$$

Question 3:

Q3) Converting ckt from to Laplace domain i.e $\left[\begin{array}{l} R \rightarrow R \\ C \rightarrow \left(\frac{C}{s}\right) \\ L \rightarrow (Ls) \end{array}\right]$

∴ the ckt is →



simplifies to



Upon doing Mesh Analysis, → [Voltage drop across closed loop = 0] KVL

① $I_1 (1+s) + (I_1 - I_2)\times\left(\frac{1}{s+1}\right) + (I_1 - I_3)(1+s) = 0$

② $I_2 (1+s) + (I_2 - I_3)\times 2 + (I_2 - I_1)\times\frac{1}{(s+1)} = 0$

③ $\cancel{-V_1(s)+} -V_1 + I_3 + (I_3 - I_1)(1+s) + (I_3 - I_2)2 + I_3\times\left(\frac{2s+3}{s+1}\right) = 0$

Upon simplification, we get →

① $I_1\left[(1+s)+\frac{1}{(s+1)}+(1+s)\right] + I_2\left[\left(\frac{-1}{s+1}\right)\right] + I_3\left[-(1+s)\right] = 0$

② $I_1\left[\frac{-1}{(1+s)}\right] + I_2\left[(1+s)+2+\frac{1}{(s+1)}\right] + I_3\left[-2\right] = 0$

③ $I_1\left[-(1+s)\right] + I_2\left[-2\right] + I_3\left[1+1+s+2+\left(\frac{2s+3}{s+1}\right)\right] = V_1$

∴ we get matrix vector form $Z(s)\, I(s) = V(s)$ as follows

$$Z(s)\, I(s) = V(s)$$

$$\Rightarrow \begin{bmatrix} 2+2s+\frac{1}{(s+1)} & \frac{-1}{1+s} & -(1+s) \\ \frac{-1}{(1+s)} & 3+s+\frac{1}{(s+1)} & -2 \\ -(1+s) & -2 & s+4+\left(\frac{2s+3}{s+1}\right) \end{bmatrix} \begin{bmatrix} I_1 \\ I_2 \\ I_3 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ V_1 \end{bmatrix}$$

$$I(s) = Z^{-1}(s)\, V(s)$$

$a_{ij} \rightarrow$ poly in s.

$$\begin{bmatrix} I_1 \\ I_2 \\ I_3 \end{bmatrix} = \begin{bmatrix} a_{00} & a_{01} & a_{02} \\ a_{10} & a_{11} & a_{12} \\ a_{20} & a_{21} & a_{22} \end{bmatrix} \begin{bmatrix} 0 \\ 0 \\ V_1 \end{bmatrix}$$

$$\Rightarrow \frac{I_1(s)}{V_1(s)} = a_{02} \quad ; \quad \frac{I_2(s)}{V_1(s)} = a_{12} \quad ; \quad \frac{I_3(s)}{V_1(s)} = a_{22}.$$

Thus, we get $I(s) = Z^{-1}(s)V(s)$ and the third column gives the required values as also depicted above.

$Z^{-1}(s)$ =

```
Z_inv =

                   2      3    4                             2     3                               2     3    4
    24 + 48s + 35s + 11s + s              9 + 13s + 7s + 2s              6 + 14s + 13s + 6s + s
  ---------------------------------  ---------------------------------  ---------------------------------
                 2       3      4   5                2       3      4   5                2       3      4   5
  57 + 144s + 147s + 74s + 17s + s  57 + 144s + 147s + 74s + 17s + s  57 + 144s + 147s + 74s + 17s + s
                    2     3                            2      3    4                          2     3
         9 + 13s + 7s + 2s               20 + 45s + 39s + 14s + s              7 + 16s + 13s + 4s
  ---------------------------------  ---------------------------------  ---------------------------------
                 2       3      4   5                2       3      4   5                2       3      4   5
  57 + 144s + 147s + 74s + 17s + s  57 + 144s + 147s + 74s + 17s + s  57 + 144s + 147s + 74s + 17s + s
                   2     3    4                            2      3                        2      3     4
      6 + 14s + 13s + 6s + s                7 + 16s + 13s + 4s            11 + 28s + 27s + 12s + 2s
  ---------------------------------  ---------------------------------  ---------------------------------
                 2       3      4   5                2       3      4   5                2       3      4   5
  57 + 144s + 147s + 74s + 17s + s  57 + 144s + 147s + 74s + 17s + s  57 + 144s + 147s + 74s + 17s + s
```

I_1(s)/V_1(s) =

$$\frac{6 + 14s + 13s^2 + 6s^3 + s^4}{57 + 144s + 147s^2 + 74s^3 + 17s^4 + s^5}$$

I_2(s)/V_1(s) =

$$\frac{7 + 16s + 13s^2 + 4s^3}{57 + 144s + 147s^2 + 74s^3 + 17s^4 + s^5}$$

I_3(s)/V_1(s) =

$$\frac{11 + 28s + 27s^2 + 12s^3 + 2s^4}{57 + 144s + 147s^2 + 74s^3 + 17s^4 + s^5}$$

Appendix: Scilab code for the complete assignment

```
// Author: Prakhar Diwan (180100083)
// Assignment for LAB-1 of EE 324

// Question 1
s = poly(0,'s'); //Defining symbol variable 's'
G1 = 10/(s^2 + 2*s + 10);
G2 = 5/(s+5);

// (a) Cascade System
//Here as per the convolution(in time domain) theorem, the overall transfer function (L) will be the product of the transfer functions of the 2 systems G1 and G2

L = G1*G2; //Transfer function of cascade system
disp("Transfer function of Cascade System", L);

// (b) Parallel System
//Here as per the superposition theorem, the overall transfer function (P) will be the sum of the transfer functions of the 2 systems G1 and G2

P = G1+G2; //Transfer function of parallel system
disp("Transfer function of Parallel System", P);

// (c) Feedback(closed loop) System
```

```
// Here using the,superposition theorem i.e. let g(t) and f(t) be 2 impulse responses with transfer functions G(s) and F(s), then (a*g(t)+b*f(t)) has a transfer function (a*G(s)+b*F(s)), where a and b are constants.
//Let R(s) be the input (in Laplace domain) and C(s) be the output(in Laplace domain) of the overall system, which has transfer function let's say T(s)
//Solving we get:
// (R(s)-(G2(s)*C(s)))*G1(s)=C(s), which simplifies to
// (C(s)/R(s))={G1(s)}/{1+(G1(s)*G2(s))}

num = G1;    //num = G1(s)
den = 1 + L; // den = 1+G1(s)*G2(s)
T = num/den; //Transfer function of Feedback(closed loop) System
disp("Transfer function of Feedback(closed loop) System", T);

// (d) Response to the unit step to the system with transfer function G1(s).
sys_G1 = syslin('c',G1);
t = linspace(0,10,1000);
y = csim('step',t,sys_G1); //output of system with transfer function G1(s) on input of unit step
plot(t,y); //plotting y(t)
title('Unit step response (in time domain) of system with transfer function G1(s)')
xlabel('Time (t)', 'fontsize',2)
ylabel('Response y(t)','fontsize',2);



// Question 2
// Poles and Zeros of the overall system in each of cases a,b,c of question 1
//Using roots on denominator and numerator of transfer function for finding poles and zeros respectively.
// 1 (a)
sys_L = syslin('c',L);
zeros_L = roots(sys_L.num);
poles_L = roots(sys_L.den);
disp("Zeros of Transfer Function", zeros_L);
disp("Poles of Transfer Function", poles_L);

// 1 (b)
sys_P = syslin('c',P);
zeros_P = roots(sys_P.num);
poles_P = roots(sys_P.den);
disp("Zeros of Transfer Function", zeros_P);
disp("Poles of Transfer Function", poles_P);

// 1 (c)
sys_T = syslin('c',T)
zeros_T = roots(sys_T.num);
poles_T = roots(sys_T.den);
disp("Zeros of Transfer Function", zeros_T);
disp("Poles of Transfer Function", poles_T);

// Computations on matrices: such as addition, multiplication,
// determinant and inverse calculation
M1 = [G1 P; L G2];
disp("M1", M1);
```

```
M2 = [T s; (1+s) G2];
disp("M2", M2);
M_add = M1 + M2;
disp("Addition of M1 and M2", M_add);
M_prod = M1*M2;
disp("Product of M1 and M2", M_prod);
M1_det = det(M1);
disp("Determinant of M1", M1_det);
M1_inv = inv(M1);
disp("Inverse of M1", M1_inv);


// Question 3
//Obtaining Z(s)I(s)=V(s) is shown in report
Z = [2+2*s+1/(1+s) -1/(1+s) -(1+s); -1/(1+s) 3+s+1/(s+1) -2; -1-s -2 6+s+1/(1+s)];
disp("Z(s)", Z);

Z_inv = inv(Z); // I(s) = Z_inv(s)*V(s)
disp("Z_inv(s)", Z_inv);

// Third column has required values result in the required quantity as depicted in report
I_1byV_1=Z_inv(1, 3);
disp("I_1/V_1", I_1byV_1)

I_2byV_1=Z_inv(2, 3)
disp("I_2/V_1", I_2byV_1)

I_3byV_1=Z_inv(3, 3)
disp("I_3/V_1", I_3byV_1)
```