EE324 Control Systems Lab

Problem Sheet 6

Prakhar Diwan, Roll no. 180100083

1. Problem 1

We are given $G(s)$, as the open loop transfer function of a system with unity negative feedback. And a proportional controller K is applied to the given system.

$$G(s) = \frac{1}{(s+3)(s+4)(s+12)}$$

The root-locus of the given system is plotted below:



Figure 1: Root-Locus for open loop transfer function G(s)

1.1. 1a

For designing a P controller to obtain a steady state error of 0.489 on applying step input, we obtain the gain value (K) as follows:

$$e(\infty) = \lim_{s \to 0} sE(s)$$
$$sE(s) = sR(s)/(1+KG(s));\ R(s) = 1/s$$

Therefore, we obtain:

$$e(\infty) = 1/(1+KG(0)) = 0.489$$

which upon solving leads to $K = (144*511)/489$.

The unit step response for the chose value of K below:



Figure 2: Unit step response for 1a

We observe that the steady state value is 0.511, which hence has an steady state error = 0.489 (1-0.511) as required.

```
clear;
clc;
s=poly(0,'s');

G = 1/((s+3)*(s+4)*(s+12));
sysG = syslin('c',G);

fig=scf();
evans(sysG);

k = ((144*511)/489);

t=0:0.01:10;
sysT = syslin('c',k*G/(1+k*G));
step = csim('step',t,sysT);
fig=scf();
plot(t,step);
xtitle ('Unit step response for steady state error = 0.489',"time instants (s)" ,
"unit step response" );
```

1.2. 1b

To obtain a damping ratio $\rho = 0.35$, we locate the intersection point of the root locus in Figure 3 and the line that makes $tan^{-1}\left(\frac{\sqrt{1-\rho^2}}{\rho}\right) = 1.2132\ radians$ with the negative real axis.

From the root locus directly, we observe that at intersection at gain = 371.9 as shown in Figure 3 with one of the pole values also visible:



Figure 3: Intersection of RL and $\zeta = 0.35$ line

Iterating over values of K with maximum error (10^{-6}) as to get a better estimate, we get $K = 371.88$.

```
Required Gain = 371.880
Poles are:
  -14.878086
  -2.0609568 + 5.5160014i
  -2.0609568 - 5.5160014i

Obtained damping factor = 0.350
```

Figure 4: At the intersection point for damping ratio = 0.35

We see that at this gain, we can approximate the system as a second order system $\{dominant\ pole\ approximation:\ as\ |Re(Far-away\ pole)| > |5{*}Re(dominant\ pole)|\}$ and hence we can define the damping ratio for the system, as equal to 0.35.
The unit step respnse for the system is plotted below:



Figure 5: Unit step response of system with damping ratio ρ = 0.35

```
clear;
clc;
s=poly(0,'s');

G = 1/((s+3)*(s+4)*(s+12));
sysG = syslin('c',G);

rho=0.35;           // Damping ratio reqd
theta=atan(sqrt(1-rho^2)/rho);   // Angle made for given rho
a=[0:0.01:10];
fig=scf();
evans(G, 500);
x=-cos(theta)*a;
y=sin(theta)*a;
plot(x, y, 'k--');
sgrid();
k_list=0.1:0.01:500;
eps=10^(-6);
D = G.den;
for k=k_list
    char_poly = D+k;
    poles=roots(char_poly);
    p_y=imag(poles(2));
    p_x=-1*real(poles(2));
    if p_x<=0 then
        break
    end
    theta_cur=atan(p_y/p_x);
    if abs(theta_cur-theta)<eps then
        printf("Required Gain = %2.3f \n", k);
        printf("Poles are: ");
        disp(poles);
        printf('Obtained damping factor = %2.3f\n', cos(theta_cur));
    end
end

k = 371.88;
t=0:0.01:10;
sysT = syslin('c',k*G/(1+k*G));
step = csim('step',t,sysT);
fig=scf();
plot(t,step);
xtitle ('Unit step response for damping ratio = 0.35',"time instants (s)" , "unit
step response" );
```

1.3. 1c

The gain value at the break away point was obtained as $K = 2.127$. It was obtained to be same from numerical calculations. The RL plot with the break away point marked is shown below:



Figure 6: Root Locus for 1c

```
clear;
clc;
s=poly(0,'s');

G = 1/((s+3)*(s+4)*(s+12));
sysG = syslin('c',G);

fig=scf();
evans(sysG,100);
```

1.4. 1d

Given the open loop system, upon increasing the controller gain K in a small range of 0.1 to 1 in steps of 0.15.
The following step responses were obtained for the above values of K:



Figure 7: Unit step responses of Closed loop system with K varying between 0 and 1

1. **Closed loop poles:** All of them lie on the open left half plane as the system reponses observed are stable. Due to the no oscillations in responses for $\forall K < 2.127$, which is the break-away point as observed in part (c), we conclude a response similar to overdamped type, which implies all poles are distinct and lie on negative real axis, for values of K between 0 and 1.
2. **Steady state errors:** Upon increase in gain value $(0 \leqslant K \leqslant 1)$, steady state value increases ($which\ is$ < 1 $\forall 0 \leqslant K \leqslant 1)$ as depicted in the above plot. and hence the steady state error with unit step input decreases as it's equal to difference of 1 and steady state value.

```
clear;
clc;
s=poly(0,'s');

G = 1/((s+3)*(s+4)*(s+12));
sysG = syslin('c',G);
t = 0 : 0.01 : 7;
k = 0.1 : 0.15 : 1;
y = zeros(length(t),length(k))
figure;
i=1;
for k1 = k
    sys = syslin('c', ((k1*G)/(k1*G+1)));
    y(:,i) = csim('step', t, sys);
    i=i+1;
end
plot(t, y)
h1=legend(string(k),4);
xlabel("t (s)")
ylabel("Step response, y(t)")
title("Step Response for k = 0.1 to 1");
f = gcf();
f.background = 8;
```

1.5. 1e

Given the open loop system, upon increasing the controller gain K in a larger range from 1 to 997 in steps of 166.
The following step responses were obtained for the above values of K:



Figure 8: Unit step responses of Closed loop system with K varying between 1 and 997

1. **Closed loop poles:** All of them lie on the open left half plane as the system reponses are stable. Due to the oscillations in responses for $\forall K > 2.127$, as $K = 2.127$ is the break-away point as observed in part (c), we conclude an underdamped type response implying 2 complex closed loop poles $\forall K > 2.127$, in the above plot and the third pole lies on real axis $\forall K$. And$\forall K \leqslant 2.127$, all the poles lie on negative real axis.
2. **Steady state errors:** Upon increase in gain value ($1 \leqslant K \leqslant 1000$), these decrease as is depicted in the above plot by increase of the steady state value towards 1.
3. **5 % settling times:** We see that as K becomes larger from 1 to the gain at the breakaway point, and the settling times decrease. However, as K increases till about 200, the settling time continues to decrease. This shows the effects of the third pole. After $K = 300$, we see that as K increases, the settling time increases, with a minima for settling time observed near $K = 600$, and then settling time increases as K increases till 1000.
   Though upon reducing the steps, I observed the variation of 5% settling time vs open loop gain K as plotted in Figure 9.
4. **Stability of the system:** As the steady state is reached for all K in the range taken. Therefore, stability of system is guaranteed for $1 \leqslant K \leqslant 1000$.



Figure 9: Variation of 5% settling time with open loop gain

```
clear;
clc;
s=poly(0,'s');

G = 1/((s+3)*(s+4)*(s+12));
sysG = syslin('c',G);
t = 0 : 0.01 : 12;
k = 1: 83 :997;
y = zeros(length(t),length(k));
figure;
i=1;
for k1 = k
    sys = syslin('c', ((k1*G)/(k1*G+1)));
    y(:,i) = csim('step', t, sys);
    i=i+1;
end
plot(t, y)
//h1=legend(string(k),8);
xlabel("t (s)")
ylabel("Step response, y(t)")
title("Step Response for k = 1 to 997");
f = gcf();
f.background = 8;
```